Computer Algebra for Physics Examples

Electrostatics, Magnetism, Circuits and Mechanics of Charged Particles
Part 2

Magnetic Field

Leon Magiera and Josef Böhm

III. Magnetic Field

A magnetic field can be described by using the following quantities: a magnetic induction vector $\vec{B}$, the magnetization vector $\vec{J}$ and magnetic strength $\vec{H}$. These quantities are related to each other through the equations

$$\vec{B} = \mu_0\left(\vec{H} + \vec{J}\right), \quad \vec{B} = \mu\mu_0\vec{H},$$

where μ_0 is the susceptibility of free space and μ is the relative susceptibility of a material medium.

When solving the problems in this chapter, we will use the following laws of physics: the laws of *Biot-Savart*, *Ampere* and *Faraday*.

According to the *Biot-Savart* law the contribution of an element of a conductor *dl* with current *I* to the magnetic field strength $\vec{H}$ at a point $\vec{r}$ is given by the formula:

$$\overrightarrow{dH} = \frac{I}{4\pi}\frac{\overrightarrow{dl} \times \vec{r}}{r^3} \qquad \text{III.1}$$

Ampere's law is of the form

$$\oint \vec{H}\, dl = I \qquad \text{III.2}$$

where the integration is along a closed line and *I* is the total current flowing through a surface bounded by the line.

Faraday's law of electromagnetic induction says that the electromotive force induced in a closed circuit is proportional to the temporal rate of change of a magnetic flux through a surface bounded by the circuit

$$E = -\frac{d\Phi}{dt} \qquad \text{III.3}$$

PROBLEMS

III.1 A magnetic field is generated in the point *P* by a straight conductor of length *l*, in which a current *I* flows. The configuration of the system is given in Fig. III.1. Assume the position of the point observed is defined by the angles α_1, α_2.



Fig III.1

Solution:

Introducing the angle α as a variable, one can perform the following substitutions in the *Biot-Savart* law:

$$x = \frac{a}{\tan\alpha}, \quad r = \frac{a}{\sin\alpha}$$

which leads to

$$dH = \frac{I}{4\pi}\frac{\sin\alpha}{r^2}dx.$$

We write the resultant magnetic field strength in form of the following integral:

$$H = \frac{I}{4\pi}\int_{\pi-\alpha_1}^{\alpha_2}\frac{\sin\alpha}{r^2}\left(\frac{dx}{d\alpha}\right)d\alpha.$$

Now we can turn to the computer algebra system:

```
(%i1) x:a/tan(alpha)$    r:a/sin(alpha)$
(%i3) H(alpha1,alpha2):=I/(4*%pi)*integrate
      (trigsimp(sin(alpha)/r^2*'diff(x,alpha)),alpha,%pi-alpha1,alpha2);
```

(%o3) $H(alpha1, alpha2) := \frac{I}{4\pi}\int_{\pi - alpha1}^{alpha2} \mathrm{trigsimp}\left(\frac{\sin(\alpha)}{r^2}\left(\frac{d}{d\alpha}x\right)\right)d\alpha$

We obtain for H:

```
(%i4) H(alpha1,alpha2);
```

(%o4) $\frac{(a\cos(alpha2) + a\cos(alpha1))\, I}{4\pi a^2}$

If point P lies on the axis of symmetry of conductor l then we have $\alpha_1 = \alpha_2$. Thus,

```
(%i5) H(alpha1,alpha1);
```

(%o5) $\frac{\cos(alpha1)\, I}{2\pi a}$

If the conductor is infinitely long, then $\alpha_1 = 0$ and we obtain the well-known result:

```
(%i6) H(0,0);
```

(%o6) $\frac{I}{2\pi a}$

The TI-NspireCAS treatment is following:

$x:=\frac{a}{\tan(\alpha)}:r:=\frac{a}{\sin(\alpha)}$	$\frac{a}{\sin(\alpha)}$
$h(\alpha 1,\alpha 2):=\frac{1}{4\cdot\pi}\cdot\int_{\pi-\alpha 1}^{\alpha 2}\left(\frac{\sin(\alpha)}{r^2}\cdot\frac{d}{d\alpha}(x)\right)d\alpha$	*Done*
$h(\alpha 1,\alpha 2)$	$\frac{\cos(\alpha 1)+\cos(\alpha 2)}{4\cdot a\cdot\pi}$
$\{h(\alpha 1,\alpha 1),h(0,0)\}$	$\left\{\frac{\cos(\alpha 1)}{2\cdot a\cdot\pi},\frac{1}{2\cdot a\cdot\pi}\right\}$

Exercise: The magnetic field strength generated by an infinitely long, straight conductor is most easily calculated by applying *Ampere's* law. Carry out the appropriate calculations.

III.2 A uniform current of surface density j flows in two infinitely long and thin tapes of width c. The distance between the tapes is b. Calculate the magnetic field strength generated in any point between the tapes and analyze the result.

Solution:

We make use of the result obtained in the previous example, in which we derived the strength of a magnetic field generated by an infinitely long conductor. Its strength is given by

$$H = \frac{I}{2\pi a},$$

where: I is the linear current, a the distance between a point and the conductor.

First we consider one tape. We define the system of coordinates as shown in Fig. III.2.



Fig III.2

One can treat the tape as an infinite number of conductors (bands of width dy) parallel to the z-axis. The value of the magnetic strength generated by one conductor (band) is given by (see previous example):

$$dH = \frac{dI}{2\pi r}, \text{ where } dI = j\,dy \text{ and } r = \sqrt{x_0^{\,2} + (y - y_0)^2}.$$

Its Cartesian coordinates are given by

$$dH_x = dH\sin(\varphi),\ dH_y = dH\cos(\varphi),\ \text{ where } \varphi = \arccos\frac{y - y_0}{x_0}.$$

The components of the resultant magnetic field can be obtained by integrating over the whole range of the y-component (width of the tape).

$$H_x = \int_{-c/2}^{c/2} dH_x,\ H_y = \int_{-c/2}^{c/2} dH_y.$$

Now let the CAS do the job:

```
(%i1) r:sqrt(x0^2+(y-y0)^2)$    phi:atan((y-y0)/x0)$
(%i3) assume(c>0)$
      H_(j,x0,y0,c):=j/(2*%pi)*
                          integrate(1/r*[sin(phi),cos(phi)],y,-c/2,c/2);
```

(%o4) $H\_(j,x0,y0,c):=\dfrac{j}{2\pi}\displaystyle\int_{\frac{-c}{2}}^{\frac{c}{2}}\frac{1}{r}[\sin(\varphi),\cos(\varphi)]\,dy$

Evaluating the integral we get:

```
(%i5) H_(j,x0,y0,c):=j/(2*%pi)*integrate(radcan(1/r*[sin(phi),cos(phi)]),y,-c/2,c/2)$
(%i6) H_(j,x0,y0,c);
```

(%o6) $\left[\dfrac{j\left(\dfrac{|x0|\log\left(\dfrac{\left|4\,x0\,y0^2-4\,c\,x0\,y0+4\,x0^3+c^2\,x0\right|}{4}\right)}{2\,x0}-\dfrac{|x0|\log\left(\dfrac{\left|4\,x0\,y0^2+4\,c\,x0\,y0+4\,x0^3+c^2\,x0\right|}{4}\right)}{2\,x0}\right)}{2\pi},\ \dfrac{j\,|x0|\left(\dfrac{\operatorname{atan}\left(\frac{2\,y0+c}{2\,x0}\right)}{x0}-\dfrac{\operatorname{atan}\left(\frac{2\,y0-c}{2\,x0}\right)}{x0}\right)}{2\pi}\right]$

Declaring the domain of x_0 and applying function `logcontract` leads to %o8.

```
(%i7) assume(x0>0)$
      logcontract(H_(j,x0,y0,c));
```

(%o8) $\left[\dfrac{j\log\left(\dfrac{\left|4\,y0^2-4\,c\,y0+4\,x0^2+c^2\right|}{\left|4\,y0^2+4\,c\,y0+4\,x0^2+c^2\right|}\right)}{4\pi},\ \dfrac{j\operatorname{atan}\left(\frac{2\,y0+c}{2\,x0}\right)-j\operatorname{atan}\left(\frac{2\,y0-c}{2\,x0}\right)}{2\pi}\right]$

We obtain the field strength for the point lying on the x-axis ($y_0 = 0$):

```
(%i9) %o8, y0=0;
```

(%o9) $\left[0,\ \dfrac{j\operatorname{atan}\left(\frac{c}{2\,x0}\right)}{\pi}\right]$

Expression %o9 indicates that the vector of the field strength his parallel to the y-axis.

We let the width of the tape tend to infinity, then we obtain:

```
(%i10) limit(%o8,c,inf);
```

(%o10) $\left[0,\ \dfrac{j}{2}\right]$

In case of two tapes the resultant field strength is the sum of the strengths generated by the individual tapes.

The point lies in-between the two tapes, $b > x_0$.

Here it is necessary to take two possibilities into account. If the currents are flowing in the same direction, then we have:

(%i11) assume(b-x0>0)$

(%i12) %o8,x0=b-x0;

$$\text{(\%o12)}\quad \left[\frac{j \log\left(\frac{\left|4\, y0^2-4\, c\, y0+4\, (b-x0)^2+c^2\right|}{\left|4\, y0^2+4\, c\, y0+4\, (b-x0)^2+c^2\right|}\right)}{4\, \pi}, \frac{j\, \mathrm{atan}\left(\frac{2\, y0+c}{2\, (b-x0)}\right)-j\, \mathrm{atan}\left(\frac{2\, y0-c}{2\, (b-x0)}\right)}{2\, \pi}\right]$$

(%i13) logcontract(%o12+%o8);

$$\text{(\%o13)}\quad \left[\frac{j \log\left(\frac{\left|4\, y0^2-4\, c\, y0+4\, x0^2+c^2\right| \left|4\, y0^2-4\, c\, y0+4\, x0^2-8\, b\, x0+c^2+4\, b^2\right|}{\left|4\, y0^2+4\, c\, y0+4\, x0^2+c^2\right| \left|4\, y0^2+4\, c\, y0+4\, x0^2-8\, b\, x0+c^2+4\, b^2\right|}\right)}{4\, \pi},\right.$$
$$\left.\frac{j\, \mathrm{atan}\left(\frac{2\, y0-c}{2\, x0-2\, b}\right)-j\, \mathrm{atan}\left(\frac{2\, y0+c}{2\, x0-2\, b}\right)}{2\, \pi}+\frac{j\, \mathrm{atan}\left(\frac{2\, y0+c}{2\, x0}\right)-j\, \mathrm{atan}\left(\frac{2\, y0-c}{2\, x0}\right)}{2\, \pi}\right]$$

Similarly, if the currents flow in opposite directions:

(%i14) %o12,j=-j;

$$\text{(\%o14)}\quad \left[-\frac{j \log\left(\frac{\left|4\, y0^2-4\, c\, y0+4\, (b-x0)^2+c^2\right|}{\left|4\, y0^2+4\, c\, y0+4\, (b-x0)^2+c^2\right|}\right)}{4\, \pi}, \frac{j\, \mathrm{atan}\left(\frac{2\, y0-c}{2\, (b-x0)}\right)-j\, \mathrm{atan}\left(\frac{2\, y0+c}{2\, (b-x0)}\right)}{2\, \pi}\right]$$

(%i15) logcontract(%o14+%o8);

$$\text{(\%o15)}\quad \left[-\frac{j \log\left(\frac{\left|4\, y0^2-4\, c\, y0+4\, x0^2-8\, b\, x0+c^2+4\, b^2\right| \left|4\, y0^2+4\, c\, y0+4\, x0^2+c^2\right|}{\left|4\, y0^2-4\, c\, y0+4\, x0^2+c^2\right| \left|4\, y0^2+4\, c\, y0+4\, x0^2-8\, b\, x0+c^2+4\, b^2\right|}\right)}{4\, \pi},\right.$$
$$\left.\frac{j\, \mathrm{atan}\left(\frac{2\, y0+c}{2\, x0-2\, b}\right)-j\, \mathrm{atan}\left(\frac{2\, y0-c}{2\, x0-2\, b}\right)}{2\, \pi}+\frac{j\, \mathrm{atan}\left(\frac{2\, y0+c}{2\, x0}\right)-j\, \mathrm{atan}\left(\frac{2\, y0-c}{2\, x0}\right)}{2\, \pi}\right]$$

The expressions obtained are general but rather complex.

We will analyze a particular case, i.e. where the tapes are of infinite width.

a) for currents flowing in the same direction

```
(%i16) limit(H_(j,x0,y0,c)+H_(j,b-x0,y0,c),c,inf);
(%o16) [0,j]
```

b) for currents flowing in opposite direction

```
(%i17) limit(H_(j,x0,y0,c)+H_(-j,b-x0,y0,c),c,inf);
(%o17) [0,0]
```

This shows that if the currents are flowing in opposite directions then the magnetic field between infinitely large tapes becomes zero.

Working with DERIVE is quite the same. There is one exception in the last two results. We cannot define $b - x_0 > 0$, so we have to substitute `SIGN(b - x0)` by 1 manually:

$$\lim_{c\to\infty} (\mathrm{H\_}(j,\ x0,\ y0,\ c) + \mathrm{H\_}(j,\ b - x0,\ y0,\ c)) = \left[0,\ \frac{j\cdot \mathrm{SIGN}(b - x0)}{2} + \frac{j}{2}\right]$$

$$\lim_{c\to\infty} (\mathrm{H\_}(j,\ x0,\ y0,\ c) + \mathrm{H\_}(-j,\ b - x0,\ y0,\ c)) = \left[0,\ \frac{j}{2} - \frac{j\cdot \mathrm{SIGN}(b - x0)}{2}\right]$$

We show the last steps performed by TI-NspireCAS and we can observe the same – very little – problem with the sign-function:

$$\left[0 \quad \frac{|2\cdot x0|}{\pi}\right]$$

$$\lim_{c\to\infty} (h\_(j,x0,y0,c))|x0>0 \qquad \left[0 \quad \frac{j}{2}\right]$$

$$h\_(j,x0,y0,c)+h\_(j,b-x0,y0,c)$$

$$\left[\frac{-\ln\left(\frac{4\cdot b^2-8\cdot b\cdot x0+c^2+4\cdot c\cdot y0+4\cdot (x0^2+y0^2)}{4\cdot b^2-8\cdot b\cdot x0+c^2-4\cdot c\cdot y0+4\cdot (x0^2+y0^2)}\right)\cdot j}{4\cdot\pi} - \frac{\ln\left(\frac{c^2+4\cdot c\cdot y0+4\cdot (x0^2+y0^2)}{c^2-4\cdot c\cdot y0+4\cdot (x0^2+y0^2)}\right)\cdot j}{4\cdot\pi} \quad \frac{\left(\tan^{-1}\left(\frac{c+2\cdot y0}{2\cdot (b-x0)}\right)+\tan^{-1}\left(\frac{c-2\cdot}{2\cdot (b-}\right.\right.}{}\right.$$

$$h\_(j,x0,y0,c)+h\_(-j,b-x0,y0,c)$$

$$\frac{\cdot (x0^2+y0^2)\big)\cdot\left(c^2-4\cdot c\cdot y0+4\cdot (x0^2+y0^2)\right)\Big)}{\cdot (x0^2+y0^2)\big)\cdot\left(c^2+4\cdot c\cdot y0+4\cdot (x0^2+y0^2)\right)\Big)}\cdot j \over 4\cdot\pi} \quad \frac{-\left(\tan^{-1}\left(\frac{c+2\cdot y0}{2\cdot (b-x0)}\right)+\tan^{-1}\left(\frac{c-2\cdot y0}{2\cdot (b-x0)}\right)-\tan^{-1}\left(\frac{c+2\cdot y0}{2\cdot x0}\right)-\tan^{-1}\left(\frac{c-2\cdot y0}{2\cdot x0}\right)\right)\cdot j}{2\cdot\pi}$$

$$\lim_{c\to\infty} (h\_(j,x0,y0,c)+h\_(j,b-x0,y0,c)) \qquad \left[0 \quad \frac{(\mathrm{sign}(b-x0)+\mathrm{sign}(x0))\cdot j}{2}\right]$$

$$\lim_{c\to\infty} (h\_(j,x0,y0,c)+h\_(-j,b-x0,y0,c)) \qquad \left[0 \quad \frac{-(\mathrm{sign}(b-x0)-\mathrm{sign}(x0))\cdot j}{2}\right]$$

III.3 Calculate the strength of a magnetic field due to a circular conductor of radius R with current I, along the axis perpendicular to the circle plane and passing through its centre (Fig II.3).

Solution:

We apply the law of *Biot-Savart* (formula III.1).

It should be noted that due to symmetry the vector of the strength of the magnetic field will be directed along the z-axis. Thus, the strength resulting from an element of the conductor $\vec{dl}$ is given by

$$\vec{dH}_z = \left|\vec{dH}\right| \sin\beta .$$



Fig III.3

The following relations hold: $r = \sqrt{R^2 + z^2}$, $\sin\beta = \dfrac{R}{r}$.

Since $\vec{dl} \perp \vec{r}$ formula III.1 reduces to

$$H_z = \frac{1}{4\pi}\int_0^{2\pi R} \frac{\sin\beta}{r^2}\,dl.$$

We enter these formulae and then evaluate the integral above:

```
(%i1) r:sqrt(R^2+z^2)$  beta:asin(R/r)$  assume (R>0)$
(%i4) Hz(z):=I/(4*%pi)*integrate(1/r^2*sin(beta),l,0,2*%pi*R);
```

(%o4) $\mathrm{Hz}(z) := \dfrac{I}{4\pi}\displaystyle\int_0^{2\pi R} \frac{1}{r^2}\sin(\beta)\,dl$

```
(%i5) hz:Hz(z);
```

(%o5) $\dfrac{I R^2}{2\,(R^2+z^2)^{3/2}}$

At the centre of the circular conductor ($z = 0$) we get:

```
(%i6) limit(hz,z,0);
```

(%o6) $\dfrac{I}{2R}$

III.4 A current I flows through a piece of arc shaped wire described by the following parameters: radius of curvature R and angle φ_0 (see Fig. III.4). Find the magnetic field strength along a line perpendicular to the plane of the arc and passing through the centre of curvature.

Solution:

The elementary contribution to the magnetic field vector at $\vec{r}$, originating from the element of current

$$I\,\overrightarrow{dl}$$

is given by the law of *Biot-Savart*

$$\overrightarrow{dH} = \frac{1}{4\pi}\frac{I\,\overrightarrow{dl}\times\vec{r}}{r^3}.$$

According to Fig. III.4 the vectors $\overrightarrow{dl}$ and $\vec{r}$ are mutually perpendicular, which allows us to simplify the formula above:

$$dH = \frac{1}{4\pi}\frac{I\,dl}{r^2}, \text{ where } dl = R\,d\varphi, r = \sqrt{R^2 + h^2}.$$



Fig III.4

The z-component of $\overrightarrow{dH}$ is

$$dH_z = dH\cos\alpha.$$

The projection of $\overrightarrow{dH}$ onto the xy-plane is equal to $dH\sin\alpha$ and its x-component is given by

$$dH_x = dH\sin\alpha\cos\varphi.$$

In the system of coordinates defined (Fig. III.4) $dH_y = 0$, due to symmetry. The resultant field vector $\vec{H}$ is obtained as result of the following integration.

$$\vec{H} = \vec{H}\left(2\int_0^{\frac{\varphi_0}{2}} dH_x, 0, 2\int_0^{\frac{\varphi_0}{2}} dH_z\right).$$

Now we can enter the expressions given above in Maxima:

Expression %o5 shows the resulting field vector.

```
(%i1) assume(R>0,h>0,phi0>0)$
(%i2) r:sqrt(R^2+h^2)$  alpha:acos(R/r)$
(%i4) H(h,phi0):=2*I/(4*%pi)*integrate(radcan(
      [sin(alpha)*cos(phi)/r^2*R,0,cos(alpha)/r^2*R]),phi,0,phi0/2);
```

$$\text{(\%o4)}\quad H(h,phi0):=\frac{2\,I}{4\,\pi}\int_0^{\frac{phi0}{2}}\operatorname{radcan}\left(\left[\frac{\sin(\alpha)\cos(\varphi)}{r^2}R,\,0,\,\frac{\cos(\alpha)}{r^2}R\right]\right)d\varphi$$

```
(%i5) H(h,phi0);
```

$$\text{(\%o5)}\quad \left[\frac{h\sin\left(\frac{phi0}{2}\right)I\,R}{2\,\pi\,(R^2+h^2)^{3/2}},\,0,\,\frac{phi0\,I\,R^2}{4\,\pi\,(R^2+h^2)^{3/2}}\right]$$

For the full circle (%o6) and then for the centre of the circle (%o7) we receive:

```
(%i6) H(h,2*%pi);
```

$$\text{(\%o6)}\quad \left[0,\,0,\,\frac{I\,R^2}{2\,(R^2+h^2)^{3/2}}\right]$$

```
(%i7) subst(h=0,%o6);
```

$$\text{(\%o7)}\quad \left[0,\,0,\,\frac{I}{2\,R}\right]$$

DERIVE and TI-NspireCAS are doing pretty similar. We don't need the assumptions. TI-Nspire doesn't distinguish between lower and upper case, so we use rr for R, etc.

$$r:=\sqrt{rr^2+h^2}:dl:=rr\cdot d\varphi:\alpha:=\cos^{-1}\left(\frac{rr}{r}\right) \qquad \cos^{-1}\left(\frac{rr}{\sqrt{h^2+rr^2}}\right)$$

$$dhh:=\frac{\frac{1}{4\cdot\pi}\cdot ii\cdot dl}{r^2} \qquad \frac{d\varphi\cdot ii\cdot rr}{4\cdot(h^2+rr^2)\cdot\pi}$$

$$dhhz:=dhh\cdot\cos(\alpha):dhhx:=dhh\cdot\sin(\alpha)\cdot\cos(\varphi)|h>0 \qquad \frac{h\cdot d\varphi\cdot ii\cdot rr\cdot\cos(\varphi)}{4\cdot(h^2+rr^2)^{\frac{3}{2}}\cdot\pi}$$

$$hh\_(h,\varphi 0):=2\cdot\int_0^{\frac{\varphi 0}{2}}\frac{\begin{bmatrix}dhhx & 0 & dhhz\end{bmatrix}}{d\varphi}\,d\varphi \qquad \text{Done}$$

$$hh\_(h,\varphi 0) \qquad \begin{bmatrix}\frac{h\cdot ii\cdot rr\cdot\sin\left(\frac{\varphi 0}{2}\right)}{2\cdot(h^2+rr^2)^{\frac{3}{2}}\cdot\pi} & 0 & \frac{ii\cdot rr^2\cdot\varphi 0}{4\cdot(h^2+rr^2)^{\frac{3}{2}}\cdot\pi}\end{bmatrix}$$

$$hh\_(0,2\cdot\pi) \qquad \begin{bmatrix}0 & 0 & \frac{ii\cdot rr^2}{2\cdot(h^2+rr^2)^{\frac{3}{2}}}\end{bmatrix}$$

For comparison, see the result of problem III.3.

III.5 We are given a solenoid of length l, with N coils of radius R. A current I flows through the coils of the solenoid. Find the formula describing the magnetic field induced along the axis of the solenoid. Find an expression for the magnitude of this field in the centre and at the ends of this axis. Calculate the magnetic field induced in these points for the following data:

$$I = 4A,\ N = 1000,\ l = 0.5m,\ R = 3{\cdot}10^{-2}m.$$

The susceptibility of free space is $\mu_0 = 4\pi \cdot 10^{-7} \dfrac{kg \cdot m}{A^2\, s^2} = 4\pi \cdot 10^{-7} \dfrac{N}{A^2}$.

Solution:

In problem III.3 we calculated the strength of the magnetic field along the axis of a circular conductor with an electric current flowing inside. Applying this result (%o5), in accordance with the notation used in Fig. III.5 …



Fig. III.5

… the strength of the magnetic field induced by an element of length dx of the solenoid in a point along the axis, which is in a distance x from this element, is given by the formula

$$dB = \frac{\mu_0\, I\, R^2}{2\left(R^2 + x^2\right)^{\frac{2}{3}}}\, n\, dx,$$

where $n = \dfrac{N}{l}$ denotes the number of coils on the solenoid per unit length. Thus, the strength of the field along the axis of the solenoid in a distance a from its centre is given by the integral expression

$$B(a) = \int_{-\frac{l}{2}+a}^{\frac{l}{2}+a} \frac{\mu_0\, I\, R^2}{2\left(R^2 + x^2\right)^{\frac{2}{3}}}\, n\, dx.$$

Now we are ready for performing the calculations:

```
(%i1) assume(l>0,R>0)$
      B(a):=%mu[0]*I*R^2*n/2*integrate(1/(R^2+x^2)^(3/2),x,-l/2+a,l/2+a);
```

$$\text{(\%o2)}\quad B(a):=\frac{\mu_0\, I\, R^2\, n}{2} \int_{\frac{-l}{2}+a}^{\frac{l}{2}+a} \frac{1}{(R^2+x^2)^{3/2}}\,dx$$

We calculate the strength in the centre of the solenoid (a = 0, gives %o4 left), at the end of it (a = l/2, gives %o4 right) and for a solenoid of infinite length (result in %o5):

```
(%i3) [B(0),B(1/2)];
```

$$\text{(\%o3)}\quad \left[\frac{\mu_0\, l\, n\, I\, R^2 \sqrt{4\,R^2+l^2}}{4\,R^4+l^2\,R^2}, \frac{\mu_0\, l\, n\, I\, R^2 \sqrt{R^2+l^2}}{2\,(R^4+l^2\,R^2)}\right]$$

```
(%i4) ratsimp(%);
```

$$\text{(\%o4)}\quad \left[\frac{\mu_0\, l\, n\, I}{\sqrt{4\,R^2+l^2}}, \frac{\mu_0\, l\, n\, I \sqrt{R^2+l^2}}{2\,R^2+2\,l^2}\right]$$

```
(%i5) limit(B(a),l,inf);
```

$$\text{(\%o5)}\quad \mu_0\, n\, I$$

Now we will carry out the numerical calculations: we calculate the magnetic induction of the induced magnetic field for the given data:

```
(%i6) %mu[0]:4*%pi*10^(-7)*kg*m/(A^2*s^2)$
      I:4*A$ n:2000/m$  R:3*10^(-2)*m$ l:1/2*m$
      kg:A*T*s^2$
      assume(m>0)$
(%i13) float([B(0),B(1/2)]);
```

$$\text{(\%o13)}\quad \left[\frac{0.009981486661471918\, kg}{s^2\, A}, \frac{0.00501752481486884\, kg}{s^2\, A}\right]$$

```
(%i14) ev(%o13);
(%o14) [0.009981486661471918 T, 0.00501752481486884 T]
```

The magnetic induction is expressed automatically in Tesla (symbol T).

We try TI-NspireCAS making use of the built-in physical constants (e.g. _µ0):

$$bb(a):=\frac{\mu0\cdot ii\cdot rr^2\cdot n}{2}\cdot\int_{\frac{-l}{2}+a} \frac{1}{\left(rr^2+x^2\right)^{\frac{3}{2}}}\,dx$$

$$\left\{bb(0),bb\left(\frac{l}{2}\right)\right\} \qquad \left\{\frac{l\cdot n\cdot ii\cdot \mu0}{\sqrt{l^2+4\cdot rr^2}},\frac{l\cdot n\cdot ii\cdot \mu0}{2\cdot\sqrt{l^2+rr^2}}\right\}$$

$$\lim_{l\to\infty}(bb(a)) \qquad n\cdot ii\cdot \mu0$$

$$\mu0:=\frac{4\cdot\pi\cdot 10^{-7}\cdot \_N}{\_A^2}:ii:=4\cdot\_A:l:=0.5\cdot\_m:rr:=3\cdot 10^{-2}\cdot\_m:n:=\frac{2000}{\_m} \qquad 2000.\cdot\frac{1}{\_m}$$

$$\left\{bb(0),bb\left(\frac{l}{2}\right)\right\} \qquad \{0.009981\cdot\_T,0.005018\cdot\_T\}$$

$$\mu0:=\_\mu0 \qquad 0.000001\cdot\frac{\_N}{\_A^2}$$

$$\left\{bb(0),bb\left(\frac{l}{2}\right)\right\} \qquad \{0.009981\cdot\_T,0.005018\cdot\_T\}$$

It is remarkable that the result is given in Tesla!

What about DERIVE?

$$\left[I := 4 \cdot A,\ N := \frac{2000}{m},\ l := 0.5 \cdot m,\ R := 3 \cdot 10^{-2} \cdot m,\ \mu 0 := \frac{4 \cdot \pi \cdot 10^{-7} \cdot kg \cdot m}{A^2 \cdot s^2} \right]$$

$$m :\in \text{Real } (0,\ \infty)$$

$$\left[B(0),\ B\left(\frac{l}{2}\right) \right] = \left[\frac{\sqrt{634} \cdot \pi \cdot kg}{7925 \cdot A \cdot s^2},\ \frac{2 \cdot \sqrt{2509} \cdot \pi \cdot kg}{62725 \cdot A \cdot s^2} \right]$$

$$\left(\text{SOLUTIONS}\left(T = \frac{kg}{A \cdot s^2},\ kg \right) \right)_1 = A \cdot T \cdot s^2$$

$$\text{SUBST}\left(\left[B(0),\ B\left(\frac{l}{2}\right) \right],\ kg,\ A \cdot T \cdot s^2 \right) = [0.009981486661 \cdot T,\ 0.005017524814 \cdot T]$$

There are also a lot of physical constants and units provided in utility files, but neither the susceptibility nor Tesla. So we have to work like with Maxima – one exception: it is not necessary to explicitly force cancelling R^2 (%i4 in Maxima).

III. 6 A conductor of length l is formed to a regular polygon with n sides. A current I flows in it. Calculate the magnetic field strength at the centre of the polygon.

Solution:

The resultant magnetic strength is the sum of the fields generated by the individual sides of the polygon.



Fig. III.6

Since the magnetic field generated by each side of the polygon is the same at the centre of the polygon, the superposition is simply

$$H(n) = n\,H_0(n)$$

where $H_0(n)$ is the field strength generated by one side in the centre of the polygon. In problem III.1 we evaluated the expression describing the strength of a magnetic field generated by a straight wire conducting current I. Using this result we have

$$H_0(n) = \frac{n\,I}{2\pi}\frac{\cos\alpha_n}{h},$$

here, however, the respective angle and distance are (see Fig. III.6):

$$\alpha_n = \frac{\pi - \dfrac{2\pi}{n}}{2} \quad \text{and} \quad h = \frac{l}{2n}\tan\alpha_n.$$

```
(%i1) alpha[n]:(%pi-2*%pi/n)/2$
      h[n]:l/(2*n)*tan(alpha[n])$
      Ho[n]:I/(2*%pi*h[n])*cos(alpha[n]);
```

$$\text{(\%o3)}\quad \frac{\cos\left(\frac{\pi - \frac{2\pi}{n}}{2}\right) n\, I}{\pi\, l \tan\left(\frac{\pi - \frac{2\pi}{n}}{2}\right)}$$

```
(%i4) H[n]:n*trigsimp(Ho[n]);
```

$$\text{(\%o4)}\quad \frac{n^2 \cos\left(\frac{\pi n - 2\pi}{2n}\right)^2 I}{\pi\, l \sin\left(\frac{\pi n - 2\pi}{2n}\right)}$$

```
(%i5) limit(H[n],n,infinity);
```

$$\text{(\%o5)}\quad \frac{\pi I}{l}$$

The field strength is given in %o4. The next expression shows the limit for $n \to \infty$, which leads to the field strength appearing in the centre of a circular conductor with circumference l. You may compare with the result of problem III.3.

DERIVE and TI-NspireCAS are behaving similar. It is a question of taste whether you prefer Maxima's %o4 or the DERIVE expression #3 for $H(n)$.

$$\#1:\quad \left[\alpha(n) := \frac{\pi - \frac{2\cdot\pi}{n}}{2},\ h := \frac{l}{2\cdot n}\cdot \mathrm{TAN}(\alpha(n))\right]$$

$$\#2:\quad H(n) := \frac{n\cdot I}{2\cdot\pi\cdot h}\cdot \mathrm{COS}(\alpha(n))$$

$$\#3:\quad H(n) = \frac{I\cdot n^2\cdot \mathrm{SIN}\left(\frac{\pi}{n}\right)^2}{\pi\cdot l\cdot \mathrm{COS}\left(\frac{\pi}{n}\right)}$$

$$\#4:\quad \lim_{n\to\infty} H(n) = \frac{\pi\cdot I}{l}$$

III. 7 A hollow portion of a sphere of radius R is electrically charged. The portion of the sphere is symmetrical with respect to the x-axis and the line from the centre of the sphere to the edge of the portion makes a maximal angle of α_{max} with the x-axis (see Fig. III.7). The surface density of the charge is constant and equal to σ. The sphere rotates with a constant angular velocity of ω about its axis of symmetry. Calculate the magnetic field strength generated in some point lying within the sphere along this axis of symmetry.

Solution:

We consider a circular element of area $dS = 2\pi r R\, d\alpha$. The charge on this element amounts to $dQ=\sigma\, dS$.



Fig. III.7

This circular element can be treated as a circular conductor with a current dI flowing through.

$$dI = \frac{dQ}{T} = \frac{dQ}{2\pi}\omega.$$

We use the solution of problem III.3.The strength of a magnetic field generated by a circular conductor of radius R with a current I flowing along the axis of symmetry in a distance z from the centre of the ring is given by

$$H = \frac{I\, R^2}{2\left(R^2 + z^2\right)^{\frac{3}{2}}}.$$

The field strength generated by the rotating ring is given by

$$dH = \frac{1}{2}\frac{r^2\, dl}{l^3},$$

where

$$r = R\sin\alpha,\;\; h = R\cos\alpha,\;\; l = \sqrt{r^2 + (h-x)^2}.$$

First we enter the relations given above. Then we perform the integration over the range of values taken by the angle α ($0 \le \alpha \le am = alphamax$)

```
(%i1) r:R*sin(alpha)$ h:R*cos(alpha)$ l:sqrt(r^2+(h-x)^2)$
      dS:2*%pi*r*R*dalpha$ dQ:sigma*dS$ dI:dQ/T$ T:2*%pi/omega$
      dH:1/2*dI*r^2/l^3$ f:1/dalpha*dH$
      assume(R>0)$
```

```
(%i11) f:ev(f,T=2*%pi/omega);
```

$$(\%o11)\quad \frac{\sin(\alpha)^3\,\omega\,\sigma\,R^4}{2\left((\cos(\alpha)\,R-x)^2+\sin(\alpha)^2\,R^2\right)^{3/2}}$$

```
(%i12) H(am):=integrate(trigsimp(f),alpha,0,am);
```

$$(\%o12)\quad H(am):=\int_0^{am}\mathrm{trigsimp}(f)\,d\alpha$$

```
(%i13) assume(am>0)$
       H(am);
```

$$(\%o14)\quad \omega\,\sigma\,R^4\left(\frac{2\,R^4-2\cos(am)\,x\,R^3+(1-\cos(am)^2)\,x^2\,R^2-2\cos(am)\,x^3\,R+2\,x^4}{6\,x^3\,R^3\sqrt{R^2-2\cos(am)\,x\,R+x^2}}-\frac{R^4-x\,R^3-x^3\,R+x^4}{3\,x^3\,R^3\sqrt{R^2-2\,x\,R+x^2}}\right.$$

Next expressions are preparations for plotting for special *alphamax*-values ($R = 4$, $\sigma = \omega = 1$):

```
(%i15) f0:subst([R=4,omega=1,sigma=1],H(%pi));
```

$$(\%o15)\quad 256\left(\frac{x^4+4\,x^3+64\,x+256}{192\,x^3\sqrt{x^2+8\,x+16}}-\frac{x^4-4\,x^3-64\,x+256}{192\,x^3\sqrt{x^2-8\,x+16}}\right)$$

```
(%i16) f1:subst([R=4,omega=1,sigma=1],H(%pi/3));
```

$$(\%o16)\quad 256\left(\frac{8\,x^4-16\,x^3+48\,x^2-256\,x+2048}{1536\,x^3\sqrt{x^2-4\,x+16}}-\frac{x^4-4\,x^3-64\,x+256}{192\,x^3\sqrt{x^2-8\,x+16}}\right)$$

```
(%i17) f2:subst([R=4,omega=1,sigma=1],H(2*%pi/3));
```

$$(\%o17)\quad 256\left(\frac{8\,x^4+16\,x^3+48\,x^2+256\,x+2048}{1536\,x^3\sqrt{x^2+4\,x+16}}-\frac{x^4-4\,x^3-64\,x+256}{192\,x^3\sqrt{x^2-8\,x+16}}\right)$$

```
(%i19) plot2d([f0,f1,f2],[x,-4,4],[y,0,4],[legend,"pi","pi/3","2pi/3"]);
```

This is the graphic representation:



For the whole sphere - $\alpha_{max} = \pi$ - we get:

```
(%i19) H(%pi);
```

$$(\%o19)\quad \omega\,\sigma\,R^4\left(\frac{R^4+x\,R^3+x^3\,R+x^4}{3\,x^3\,R^3\sqrt{R^2+2\,x\,R+x^2}}-\frac{R^4-x\,R^3-x^3\,R+x^4}{3\,x^3\,R^3\sqrt{R^2-2\,x\,R+x^2}}\right)$$

In order to show that field strength within the sphere does not depend on the position of a point we calculate

```
(%i20) radcan(H(%pi));
```

$$(\%o20)\quad \frac{2\,\omega\,\sigma\,R}{3}$$

It can be seen from %o20 that the magnetic field strength remains constant within the sphere. This can be concluded by inspecting the graphic representation, too (blue function graph).

Let me present the functions with DERIVE and the TI-NspireCAS graphs below:

$$\left[\begin{array}{c} \dfrac{4\cdot(64-x^3)\cdot\mathrm{SIGN}(x-4)}{3\cdot x^3}+\dfrac{4\cdot(x^4-2\cdot x^3+6\cdot x^2-32\cdot x+256)}{3\cdot x^3\cdot\sqrt{(x^2-4\cdot x+16)}} \\[2ex] \dfrac{4\cdot(64-x^3)\cdot\mathrm{SIGN}(x-4)}{3\cdot x^3}+\dfrac{4\cdot(x^4+2\cdot x^3+6\cdot x^2+32\cdot x+256)}{3\cdot x^3\cdot\sqrt{(x^2+4\cdot x+16)}} \\[2ex] \dfrac{4\cdot(64-x^3)\cdot\mathrm{SIGN}(x-4)}{3\cdot x^3}+\dfrac{4\cdot(x^3+64)\cdot\mathrm{SIGN}(x+4)}{3\cdot x^3} \end{array}\right]$$



III. 8 A conductor is fixed at the points A and B (Fig.III.8) in a rectangular frame with sides of length a and b. The frame lies in a uniform magnetic field $\boldsymbol{B}$**,** which is perpendicular to the plane of the frame and which increases linearly with time ($B = k\ t$.) . The resistance of a unit length of the conductor is r. Calculate the current induced in the conductor and the electrical potential difference between the points A and B.

Solution:

In the circuits ABD, ABC (Fig.III.8)



Fig. III. 8

the electromotive forces induced are given by

$$E_1 = -\frac{d\Phi_1}{dt} \quad and \quad E_2 = -\frac{d\Phi_2}{dt},$$

where Φ_1, Φ_2 denote the fluxes of the vector of magnetic induction through planes ABD and ABC, respectively. These can be written in the form

$$\Phi_1 = BS_1,\ \Phi_2 = BS_2,$$

where the surface areas S_1, S_2 of the circuits are given by

$$S_1 = a\ (b - c),\ S_2 = a\ c.$$

We remember the *Kirchhoff* laws and we turn to our CAS tool:

(This problem was treated by using the latest wxMaxima-version 15.08.1 which provides among other improvements Greek characters and mathematical symbols.)

```
(%i7)   B:k*t$  Φ[1]:B*S[1]$  Φ[2]:B*S[2]$
        S[1]:a*(b-c)$  S[2]:a*c$
        E[1]:-diff(Φ[1],t)$  E[2]:-diff(Φ[2],t)$
(%i10)  eq1:I1*r*(a+2*(b-c))+I3*r*a=E[1]$
        eq2:I2*r*(a+2*c)-I3*r*a=E[2]$
        eq3:I3=I1-I2$
```

We calculate the currents by solving the above simultaneous equations

```
(%i11) soln:solve([eq1,eq2,eq3],[I1,I2,I3])[1];
```

(soln) $$\left[I1=-\frac{\left(2\,a\,c^2+(a^2-2\,a\,b)\,c-2\,a^2\,b\right)k}{(4\,c^2-4\,b\,c-4\,a\,b-3\,a^2)\,r},\ I2=-\frac{\left(2\,a\,c^2+(-2\,a\,b-a^2)\,c-a^2\,b\right)k}{(4\,c^2-4\,b\,c-4\,a\,b-3\,a^2)\,r},\ I3=-\frac{(2\,a^2\,c-a^2\,b)\,k}{(4\,c^2-4\,b\,c-4\,a\,b-3\,a^2)\,r}\right]$$

It should be noticed that the direction of current I_3 is in accordance with the direction indicated in the figure if $c > \frac{b}{2}$.

Next we calculate the potential drop along the conductor AB:

```
(%i13)   I3:rhs(soln[3])$
         UAB:I3*c*r$
(%i14)   ev(UAB);
```

$$(\%o14)\quad -\frac{c\,(2\,a^2\,c - a^2\,b)\,k}{4\,c^2 - 4\,b\,c - 4\,a\,b - 3\,a^2}$$

It can easily be seen that the potential drop *UAB* and consequently current I_3 is equal zero for $c = \frac{b}{2}$. We have:

```
(%i15)   solve(%=0,c);
```

$$(\%o15)\quad [\,c = \frac{b}{2}\,,\ c = 0\,]$$

From the physical point of view, however, the only satisfactory solution is $c = \frac{b}{2}$.

■

III. 9 A thin conducting ring is placed in a uniform magnetic field, whose induction *B* fluctuates according to the formula $B = B_0 \cos(\omega\, t)$. The radius of the ring is *r,* its resistance *R*, and coefficient of self-inductance *L*. The induction vector $\overline{B}$ lies at an angle α to the plane of the ring. Calculate the average moment of the forces acting upon the ring.

Solution:

The torque $\overline{M}$ acting upon a flat circuit placed in a uniform magnetic field of induction $\vec{B}$, in which a current *I* flows, is given by the formula

$$\overline{M} = I\,S\,(\vec{n} \times \vec{B}),$$

where $\vec{n}$ denotes a unit vector perpendicular to the plane *S* containing the conducting circuit.



Fig.III.9

In accordance with Fig.III.9, the instantaneous value of the moment of force is given by

$$M = IBS\sin\left(\frac{\pi}{2} - \alpha\right) = IBS\cos(\alpha)$$

From *Ohm*'s law the current flowing in the circuit is directly proportional to the electromotive force E

$$E = I\,R$$

The resultant electromotive force E is equal to the sum of the electromotive forces induced as a result of the changes in the flux of the induction of the external magnetic field and the electromotive force of self-induction

$$E = -\frac{d\Phi}{dt} - L\frac{dI}{dt}.$$

Using the notation used in Fig.III.7, the flux of the magnetic induction vector through the plane of the ring is given by

$$\Phi = B\,S\cos\left(\frac{\pi}{2} - \alpha\right) = B\,S\sin\alpha,\ \text{ where }\ S = \pi\,r^2.$$

We start entering the given relations:

```
(%i4)  B(t):=Bo*cos(ω*t)$  Φ(t):=B(t)*S*sin(α)$
       S:%pi*r^2$  E(t):=I(t)*R$
```

The differential equation resulting from *Ohm*'s law has the form:

```
(%i5)  de:E(t)=-diff(Φ(t),t)-L*diff(I(t),t);
```

$$\text{(de)}\quad I(t)\,R = \pi\,Bo\,r^2\sin(\alpha)\,\omega\sin(t\,\omega) - \left(\frac{d}{d\,t}\,I(t)\right)L$$

In order to find current $I(t)$ we use the procedure `desolve` for solving the differential equation:

```
(%i7)  assume(R>0,ω>0,L>0)$
       gsoln:desolve(de,I(t));
```

$$\text{(gsoln)}\quad I(t) = \frac{\pi\,Bo\,r^2\,R\sin(\alpha)\,\omega\sin(t\,\omega)}{L^2\,\omega^2+R^2} - \frac{\pi\,Bo\,r^2\,L\sin(\alpha)\,\omega^2\cos(t\,\omega)}{L^2\,\omega^2+R^2} + \frac{\%e^{-\frac{t\,R}{L}}\left(\left(\pi\,Bo\,r^2\,L^2\sin(\alpha)+I(0)\,L^3\right)\omega^2+I(0)\,L\,R^2\right)}{L\left(L^2\,\omega^2+R^2\right)}$$

```
(%i8)  I(t):=expand(rhs(gsoln))$
```

```
(%i9)  I(t);
```

$$\text{(\%o9)}\quad \frac{\pi\,Bo\,r^2\,R\sin(\alpha)\,\omega\sin(t\,\omega)}{L^2\,\omega^2+R^2} - \frac{\pi\,Bo\,r^2\,L\sin(\alpha)\,\omega^2\cos(t\,\omega)}{L^2\,\omega^2+R^2} + \frac{\pi\,Bo\,r^2\,L^2\sin(\alpha)\,\omega^2}{L^3\,\%e^{\frac{t\,R}{L}}\,\omega^2+L\,R^2\,\%e^{\frac{t\,R}{L}}} + \frac{I(0)\,L^3\,\omega^2}{L^3\,\%e^{\frac{t\,R}{L}}\,\omega^2+L\,R^2\,\%e^{\frac{t\,R}{L}}} + \frac{I(0)\,L\,R^2}{L^3\,\%e^{\frac{t\,R}{L}}\,\omega^2+L\,R^2\,\%e^{\frac{t\,R}{L}}}$$

Right hand side of the solution (%o9) describes how the current varies in time. It can easily be seen that three terms of the sum tend to zero when time tends to infinity:

```
(%i10)  limit([third(I(t)),fourth(I(t)),fifth(I(t))],t,inf);
(%o10)  [0,0,0]
```

We are interested in the remaining terms describing the steady state of the system:

```
(%i11)  II(t):=first(I(t))+second(I(t))$
(%i12)  II(t);
```

$$(\%o12)\quad \frac{\pi\, Bo\, r^2 R \sin(\alpha)\, \omega \sin(t\,\omega)}{L^2 \omega^2 + R^2} - \frac{\pi\, Bo\, r^2 L \sin(\alpha)\, \omega^2 \cos(t\,\omega)}{L^2 \omega^2 + R^2}$$

In the next step we enter the moment of the force $M(t)$ and then proceed finding the requested average moment of the force *Mav*.

```
(%i13)  M(t):=II(t)*S*B(t)*sin(%pi/2-α)$
(%i15)  T:2*%pi/ω$
        Mav:factor(1/T*integrate(M(t),t,0,T));
```

$$(Mav)\quad -\frac{\pi^2 Bo^2 r^4 L \cos(\alpha) \sin(\alpha)\, \omega^2}{2\,(L^2 \omega^2 + R^2)}$$

The screen shot below shows how to solve the problem with TI-NspireCAS.

$$ee:=rr\cdot ii(t) \qquad ii(t)\cdot rr$$

$$bb0\cdot \sin(\alpha)\cdot \omega\cdot \pi\cdot \sin(\omega\cdot t)\cdot r^2-\frac{d}{dt}(ii(t))\cdot ll=ii(t)\cdot rr \qquad bb0\cdot \sin(\alpha)\cdot \omega\cdot \pi\cdot \sin(\omega\cdot t)\cdot r^2-\frac{d}{dt}(ii(t))\cdot ll=ii(t)\cdot rr$$

$$\text{deSolve}\left(bb0\cdot \sin(\alpha)\cdot \omega\cdot \pi\cdot \sin(\omega\cdot t)\cdot r^2-jj'\cdot ll=jj\cdot rr,t,jj\right)$$

$$jj=r^2\cdot\left(\frac{bb0\cdot rr\cdot \sin(\alpha)\cdot \omega\cdot \pi\cdot \sin(\omega\cdot t)}{ll^2\cdot \omega^2+rr^2}-\frac{bb0\cdot ll\cdot \sin(\alpha)\cdot \omega^2\cdot \pi\cdot \cos(\omega\cdot t)}{ll^2\cdot \omega^2+rr^2}\right)+c1\cdot e^{\frac{-rr\cdot t}{ll}}$$

$$ii(t):=r^2\cdot\left(\frac{bb0\cdot rr\cdot \sin(\alpha)\cdot \omega\cdot \pi\cdot \sin(\omega\cdot t)}{ll^2\cdot \omega^2+rr^2}-\frac{bb0\cdot ll\cdot \sin(\alpha)\cdot \omega^2\cdot \pi\cdot \cos(\omega\cdot t)}{ll^2\cdot \omega^2+rr^2}\right)+c1\cdot e^{\frac{-rr\cdot t}{ll}}\Big|c1=0 \qquad \text{Done}$$

$$mm:=ii(t)\cdot ss\cdot bb\cdot \sin\left(\frac{\pi}{2}-\alpha\right) \qquad \frac{-bb0^2\cdot \sin(\alpha)\cdot \cos(\alpha)\cdot \omega\cdot \pi^2\cdot r^4\cdot \cos(\omega\cdot t)\cdot(ll\cdot \omega\cdot \cos(\omega\cdot t)-rr\cdot \sin(\omega\cdot t))}{ll^2\cdot \omega^2+rr^2}$$

$$\frac{1}{tt}\cdot\int_0^{tt} mm\, dt\Big|tt=\frac{2\cdot \pi}{\omega} \qquad \frac{-bb0^2\cdot ll\cdot \sin(\alpha)\cdot \cos(\alpha)\cdot \omega^2\cdot \pi^2\cdot r^4}{2\cdot(ll^2\cdot \omega^2+rr^2)}$$

III. 10 A uniform rod of length of length l and mass m is placed on two parallel, horizontal rails. These rails are connected to a source of constant potential difference U and placed in a constant magnetic field of strength B. This magnetic field is perpendicular to the plane containing the rails. The coefficient between the rod and the rails equals μ.

a) What is the velocity of the rod at time t?

b) Assuming that the rails are infinitely long, calculate the maximum velocity of the rod.

Solution:



Fig. III 10

The motion of the rod is determined by a force $\vec{F}$ which is the resultant of two forces: the electro-magnetic force $\overrightarrow{F_{el}}$ and the force of friction $\overrightarrow{F_f}$.

$$F = F_{el} - F_f \quad \text{where} \quad F_{el} = B\,I\,l \quad \text{and} \quad F_f = \mu G = \mu m g.$$

Motion of the rod causes a change in the flux of the magnetic field Φ and an electromotive force is induced

$$E = -\frac{d\Phi}{dt}, \quad \text{where} \quad \Phi = B\,S = B\,l\,x(t).$$

The current flowing in the circuit is given according to *Ohm*'s law by

$$I = \frac{U+E}{R}.$$

Now we can enter all above given relations

```
(%i7)   Φ(t):=B*l*x(t)$   E(t):=-diff(Φ(t),t)$
        I(t):=(U+E(t))/R$   F[el](t):=B*I(t)*l$
        F[f]:μ*G$   G:m*g$   F(t):=F[el](t)-F[f]$
```

and we apply *Newton*'s 2nd law (Force = Mass times Acceleration):

```
(%i8)   eq:m*diff(x(t),t,2)=F(t);
```

$$\text{(eq)} \qquad m\left(\frac{d^2}{d\,t^2}\,x(t)\right) = \frac{l\,B\left(U - l\left(\frac{d}{d\,t}\,x(t)\right)B\right)}{R} - G\,\mu$$

Two ways of calculating the requested velocity of the rod i.e. the unknown function $x'(t)$ are presented below.

Method 1: We separate the variables in equation `(eq)` and integrate wrt t.

```
(%i9)  eq/rhs(eq);
```

$$\text{(\%o9)}\quad \frac{m\left(\frac{d^2}{d\,t^2}\,x(t)\right)}{\frac{l\,B\left(U-l\left(\frac{d}{d\,t}\,x(t)\right)B\right)}{R}-G\,\mu}=1$$

```
(%i10) integrate(%,t);
```

$$\text{(\%o10)}\quad -\frac{m\,R\log\left(\frac{l\,B\left(U-l\left(\frac{d}{d\,t}\,x(t)\right)B\right)}{R}-G\,\mu\right)}{l^2\,B^2}=t+\%c1$$

Then we solve the resulting equation for $x'(t)$ …

```
(%i11) gsoln:solve(%,diff(x(t),t));
```

$$\text{(gsoln)}\quad \left[\frac{d}{d\,t}\,x(t)=-\frac{G\,R\,\mu-l\,B\,U+R\,\%e^{-\frac{l^2\,t\,B^2}{m\,R}-\frac{\%c1\,l^2\,B^2}{m\,R}}}{l^2\,B^2}\right]$$

… and extract the right hand side which represents the velocity.

```
(%i12) v(t):=rhs(gsoln[1])$
(%i13) v(t);
```

$$\text{(\%o13)}\quad -\frac{G\,R\,\mu-l\,B\,U+R\,\%e^{-\frac{l^2\,t\,B^2}{m\,R}-\frac{\%c1\,l^2\,B^2}{m\,R}}}{l^2\,B^2}$$

When time tends to infinite we get maximal velocity. This fact can formally be confirmed by calculation of the limit of the velocity.

```
(%i14) solve(diff(v(t),t)=0,t);
(%o14) []
(%i16) assume(l>0,m>0,R>0)$
       limit(v(t),t,inf);
Is  B  zero or nonzero?n;
```

$$\text{(\%o16)}\quad -\frac{G\,R\,\mu-l\,B\,U}{l^2\,B^2}$$

I'd like to give an additional graphic confirmation by entering numerical data and then plotting the velocity function:

```
(%i24) G:5$ R:20$ μ:0.03$
       B:1$ U:2$ l:10$ m:1$ %c1:0$
(%i25) ev(%o16);
(%o25) 0.17
```

```
(%i26) ev(%o13);
```

$$(\%o26)\quad -\frac{20\,\%e^{-5\,t}-17.0}{100}$$

```
(%i28) plot2d(-(20*%e^(-5*t)-17)/100,[t,0,2]);
```



Method 2: We choose applying function `ode2`.

We enter equation `(eq)` from above directly and solve the differential equation.

```
(%i2) de:m*(diff(v(t),t))=(l*B*(U-l*v(t)*B))/R-μ*G;
```

$$(de)\quad m\left(\frac{d}{d\,t}\,v(t)\right)=\frac{l\,B\,(U-l\,v(t)\,B)}{R}-G\,\mu$$

```
(%i3) gsoln:ode2(de,v(t),t)$
(%i4) v(t):=rhs(gsoln)$
(%i5) v(t);
```

$$(\%o5)\quad \%e^{-\frac{l^2\,t\,B^2}{m\,R}}\left(\%c-\frac{\%e^{\frac{l^2\,t\,B^2}{m\,R}}\,(G\,R\,\mu-l\,B\,U)}{l^2\,B^2}\right)$$

```
(%i6) limit(%o5,t,inf);
Is l zero or nonzero?n;
  Is m positive or negative?p;
  Is B zero or nonzero?n;
  Is R positive or negative?p;
```

$$(\%o6)\quad -\frac{G\,R\,\mu-l\,B\,U}{l^2\,B^2}$$

It is charming to compare solving the ODE applying DERIVE and TI-NspireCAS as well.

In DERIVE we make use of built-in `DSOLVE1(p,q,t,y,t0,v0)` after rewriting the equation in the form $p(t,v) + q(t,v)\cdot v' = 0$, $v(t_0) = v_0$.

#1: CaseMode := Sensitive

#2: $v\_(t) := \left(\text{SOLUTIONS}\left(\text{DSOLVE1}\left(-\dfrac{l\cdot B\cdot(U - l\cdot v\cdot B)}{R} + \mu\cdot G,\ m,\ t,\ v,\ 0,\ 0\right),\ v\right)\right)_1$

#3: $v\_(t) := \dfrac{e^{-B^2\cdot l^2\cdot t/(R\cdot m)}\cdot(G\cdot R\cdot\mu - B\cdot U\cdot l)}{B^2\cdot l^2} + \dfrac{B\cdot U\cdot l - G\cdot R\cdot\mu}{B^2\cdot l^2}$

#4: [R :∈ Real (0, ∞), m :∈ Real (0, ∞)]

#5: $\lim\limits_{t\to\infty} v\_(t) = \dfrac{B\cdot U\cdot l - G\cdot R\cdot\mu}{B^2\cdot l^2}$

TI-NspireCAS makes entering the equation easier but here it is not possible to find the limit for *t* tending to infinity. But we can see this by inspecting the exponential expression, of course!

$\text{deSolve}\left(m\cdot v'=\dfrac{l\cdot bb\cdot(uu-l\cdot v\cdot bb)}{rr}-\mu\cdot gg,t,v\right)$ $\quad v=c3\cdot e^{\frac{-l^2\cdot bb^2\cdot t}{m\cdot rr}}+\dfrac{l\cdot bb\cdot uu-gg\cdot rr\cdot\mu}{l^2\cdot bb^2}$

$v\_(t):=\text{right}\left(\text{deSolve}\left(m\cdot v'=\dfrac{l\cdot bb\cdot(uu-l\cdot v\cdot bb)}{rr}-\mu\cdot gg \text{ and } v(0)=0,t,v\right)\right)$ $\quad$ *Done*

$v\_(t)$ $\quad \dfrac{l\cdot bb\cdot uu-gg\cdot rr\cdot\mu}{l^2\cdot bb^2}-\dfrac{(l\cdot bb\cdot uu-gg\cdot rr\cdot\mu)\cdot e^{\frac{-l^2\cdot bb^2\cdot t}{m\cdot rr}}}{l^2\cdot bb^2}$

$\lim\limits_{t\to\infty}(v\_(t))$ $\quad$ undef

$\lim\limits_{t\to\infty}(v\_(t))|rr>0 \text{ and } m>0$ $\quad$ undef

Problem III.11 is similar.

III. 11 Two long vertical rails are closed at the upper end by a resistance R. A conductor of mass m and length l falls without friction along the rails. The whole system is placed in a uniform magnetic field of induction $\vec{B}$ perpendicular to the plane of the system.

Calculate the velocity of the falling conductor as a function of time.

Solution:

The system of the rails and the conductor creates an electric circuit.

The fall of the conductor causes an electromotive force

$$E = -\frac{d\Phi}{dt} \quad \text{with } (\Phi(t) = B \cdot l \cdot x(t)).$$

The current flowing in the circuit of resistance R is given by

$$I = \frac{E}{R}.$$

Within the magnetic field, an electro-dynamic force F_{el} and a gravitational force $m\, g$ act upon the conductor with the current flowing in it. Vector $\vec{B}$ is perpendicular to the conductor. Thus the electro-dynamic force is given by (see Fig. III.11)

$$F_{el} = B \cdot I \cdot l.$$



Fig. III 11

```
(%i4)    Φ(t):=B*l*x(t)$   E(t):=-diff(Φ(t),t)$
         I(t):=E(t)/R$   F[el](t):=B*I(t)*l$
```

The electro-dynamic force counteracts the changes in the magnetic flux, thus it is directed vertically upwards. Hence, we obtain the equation for the dynamics of the system:

```
(%i5)    eq:m*diff(v(t),t)=m*g+F[el](t);
```

$$\text{(eq)} \quad m\left(\frac{d}{dt}\, v(t)\right) = g\, m - \frac{l^2 \left(\frac{d}{dt}\, x(t)\right) B^2}{R}$$

Substituting $x'(t)$ by $v(t)$, we obtain the differential equation (%o6) which can be successfully solved:

```
(%i6)   eq,diff(x(t),t)=v(t);
```

(%o6) $$m\left(\frac{d}{d\,t}\,v(t)\right)=g\,m-\frac{l^2\,v(t)\,B^2}{R}$$

```
(%i7)   soln:(ode2(%,v(t),t))$
(%i8)   v(t):=rhs(soln)$
(%i9)   v(t);
```

(%o9) $$\%e^{-\frac{l^2\,t\,B^2}{m\,R}}\left(\frac{g\,m\,R\,\%e^{\frac{l^2\,t\,B^2}{m\,R}}}{l^2\,B^2}+\%c\right)$$

In the last step the velocity for sufficiently large t is evaluated:

```
(%i11)  assume(l>0,m>0,R>0)$
        limit(v(t),t,inf);
Is  B  zero  or  nonzero?n;
```

(%o11) $$\frac{g\,m\,R}{l^2\,B^2}$$

Exercise: Expression %o6 "invites" to separate the variables and then to solve the ODE. You might try to do this manually without technology support.